
AMPL: introduzione all’uso avanzato (II parte)

La lezione del 31/10/14 è stata dedicata all’uso avanzato di insiemi ed
espressioni

Nuovi argomenti avanzati:

1 Parametri

2 Gestione dei dati

Parametri randomici

AMPL permette di generare parametri randomici in base a diverse
funzioni di distribuzione

Esempio di dichiarazione (file .mod):

param media ;

param varianza ;

param p := Normal(media, varianza) ;

Le distribuzioni sono implementate utilizzando un generatore di
numeri pseudocasuali, a partire da un certo seme:

options randseed; # per vedere il seme

options randseed val; # per cambiare il seme

Principali distribuzioni

Funzione Distribuzione

Beta(a,b) distribuzione beta con parametri a e b
Cauchy() distribuzione di Cauchy
Exponential() distribuzione esponenziale
Gamma(a) distribuzione Gamma con parametro a
Normal(m,v) distribuzione normale con media m e varianza v
Normal01() distribuzione normale con media 0 e varianza 1
Poisson(m) distribuzione di Poisson con parametro a
Uniform(a,b) distribuzione uniforme sull’intervallo [a, b]
Uniform01() distribuzione uniforme sull’intervallo [0, 1]

Parametri simbolici

Alcuni parametri possono rappresentare stringhe di caratteri.

param p symbolic; # nel file.mod

param p:=’stringa’; nel file.dat

Per selezionare elementi particolari di un insieme:

#nel file.mod

set INS; param elemento1 symbolic in INS;

#nel file.dat

set INS=a b c d; param elemento1:=d;

Per associare stringhe descrittive agli elementi di un insieme:

#nel file.mod

set mesi:={1..12};

param nomemesi{mesi} symbolic;

#nel file.dat

param mesi:=1 "gennaio" ... 12 "dicembre";

Dati: specificare insiemi multidimensionali

AMPL permette di definire gli insiemi del modello (file .mod) in modo
astratto rispetto alle strutture dati per rappresentarli

In generale, le strutture dati sono popolate mediante l’assegnamento
di opportune liste di valori (file .dat)

#file .mod

set ORIG; # origins

set DEST; # destinations

set LINKS within {ORIG,DEST}; # transportation links

#file .dat

set ORIG := GARY CLEV PITT ;

set DEST := FRA DET LAN WIN STL FRE LAF ;

set LINKS :=

(GARY,DET) (GARY,LAN) (GARY,STL) (GARY,LAF) (CLEV,FRA)

(CLEV,DET) (CLEV,LAN) (CLEV,WIN) (CLEV,STL) (CLEV,LAF)

(PITT,FRA) (PITT,WIN) (PITT,STL) (PITT,FRE) ;

Data template (*)

AMPL consente rappresentazioni alternative mediante l’uso del data
template *

Il data template consente una rappresentazione compatta dei valori di
una lista

Data una sequenza di valori, il simbolo “*” permette di generare un
valore per ogni elemento della sequenza associata ad esso

set LINKS :=

(GARY,*) DET LAN STL LAF

(CLEV,*) FRA DET LAN WIN STL LAF

(PITT,*) FRA WIN STL FRE ;

Il simbolo “*” ha rilevanza posizionale in un’espressione

Dati: specificare parametri multidimensionali

Il data template può essere usato per specificare i parametri definiti su
insiemi multidimensionali

#file .mod

param cost {LINKS} >= 0;

#file .dat

param cost :=

[GARY,*] DET 14 LAN 11 STL 16 LAF 8

[CLEV,*] FRA 27 DET 9 LAN 12 WIN 9 STL 26 LAF 17

[PITT,*] FRA 24 WIN 13 STL 28 FRE 99 ;

Rappresentazione tabellare

Le tabelle possono essere usate per rappresentare parametri associati
a insiemi multidimensionali

Il simbolo “.” è un segnaposto per le coppie riportate in tabella che
non sono definite nell’insieme

param cost: FRA DET LAN WIN STL FRE LAF :=

GARY . 14 11 . 16 . 8

CLEV 27 9 12 9 26 . 17

PITT 24 . . 13 28 99 . ;

Il tradeoff tra lista supportata da data template e forma tabellare è
dato dalla sparsità dell’insieme

Altri esempi sul data template 1/2

#file .mod

set PROD;

set ROUTES within {ORIG,DEST,PROD};

#file .dat

set PROD := bands coils ;

set ROUTES :=

(*,FRA,*) CLEV bands CLEV coils PITT bands

(*,DET,*) CLEV bands CLEV coils

(*,LAN,*) GARY coils CLEV bands CLEV coils

(*,WIN,*) CLEV coils PITT bands

(*,STL,*) GARY coils CLEV bands CLEV coils PITT bands

(*,FRE,*) PITT bands PITT coils

(*,LAF,*) GARY coils CLEV bands ;

Altri esempi sul data template 1/2

#file .mod

param cost {ROUTES} >= 0;

#file .dat

param cost :=

[*,*,bands] CLEV FRA 27 CLEV DET 9 CLEV LAN 12

CLEV STL 26 CLEV LAF 17 PITT FRA 24

PITT WIN 13 PITT STL 28 PITT FRE 99

[*,*,coils] GARY LAN 11 GARY STL 16 GARY LAF 8

CLEV FRA 23 CLEV DET 8 CLEV LAN 10

CLEV WIN 9 CLEV STL 21 PITT FRE 81 ;

G. Leo (UniRoma2) AMPL & CPLEX 21/11/14 11 / 13



Tabelle multidimensionali

Una tabella m-dimensionale T con m ≥ 3 dimensioni può essere espressa
mediante una tabella bidimensionale B tale che:

una dimensione di B corrisponde a m − 1 insieme aggregati di T
l’altra dimensione di B corrisponde al restante insieme di T

param cost: bands coils :=

CLEV FRA 27 23

CLEV DET 8 8

CLEV LAN 12 10

CLEV WIN . 9

CLEV STL 26 21

CLEV LAF 17 .

PITT FRA 24 .

PITT WIN 13 .

PITT STL 28 .

PITT FRE 99 81

GARY LAN . 11

GARY STL . 16

GARY LAF . 8 ;

Data template tabelle multidimensionali

Il data template per le tabelle è espresso da due simboli:
“*” rappresenta un indice sulle righe;
“:” rappresenta un indice sulle colonne;

Il data template è dato da una sequenza ordinata di simboli con
separatore: “,” e delimitatori: “[ ]”
L’ordinamento dei simboli rispetta l’ordine degli insiemi nella
definizione del parametro (file .mod)

#def (file .mod) --> set ROUTES within {ORIG,DEST,PROD};

#def (file .mod) --> param cost {ROUTES} >= 0;

param cost :=

[*,:,*]: FRA DET LAN WIN STL FRE LAF :=

CLEV bands 27 9 12 . 26 . 17

CLEV coils 23 8 10 9 21 . .

PITT bands 24 . . 13 28 99 .

PITT coils . . . . . 81 .

GARY coils . . 11 . 16 . 8 ;

